9/20

11.2.3 Constant-Strength Vortex Method

The constant-strength vortex distribution was shown to be equivalent to a linear-strength doublet distribution (Section 10.3.2) and therefore is expected to improve the solution of the flow over thick bodies. However, this method is more difficult to use successfully compared to the other methods presented here. One of the problems arises from the fact that the self-induced effect (Eq. (10.43)) of this panel is zero at the center of the element (and the influence coefficient matrix, without a pivoting scheme, will have a zero diagonal). Also, when using the Kutta condition at an airfoil's trailing edge (Fig. 11.23) the requirement that $\gamma_1 + \gamma_N = 0$ eliminates the lift of the two trailing edge panels. Consequently, if N panels are used, then only $N-2$ independent equations can be used and the scheme cannot work without certain modifications to the method. One such modification is presented in Ref. 5.1 (pp. 281–282) where additional conditions are found by minimizing a certain error function. In this section, we try to use an approach similar to the previous source and doublet methods, and only the specifications of the boundary conditions will be modified. We will follow the basic six-step procedure of the previous sections.

Selection of singularity element. Consider the constant-strength vortex element of Section 10.2.3, where the panel is based on a flat surface element. To establish a normal-velocity boundary condition based method, only the induced velocity formulas are used (Eqs. (10.39) and (10.40)):

$$u_p = \frac{\gamma}{2\pi}\left[\tan^{-1}\frac{z-z_2}{x-x_2} - \tan^{-1}\frac{z-z_1}{x-x_1}\right] \qquad \text{panel coordinates} \qquad (11.44)$$

$$w_p = -\frac{\gamma}{4\pi}\ln\frac{(x-x_1)^2+(z-z_1)^2}{(x-x_2)^2+(z-z_2)^2} \qquad \text{panel coordinates} \qquad (11.45)$$



FIGURE 11.23
Constant-strength vorticity panels near the trailing edge of an airfoil.

Here, again, the velocity components $(u, w)_p$ are in the direction of the panel local coordinates, which need to be transformed back to the x–z system by Eq. (11.23).

This procedure can be included in an induced-velocity subroutine VOR2DC (where C stands for constant) that will compute the velocity (u, w) at an arbitrary point (x, z) due to the jth element:

$$(u, w) = \text{VOR2DC}\,(\gamma_j, x, z, x_j, z_j, x_{j+1}, z_{j+1}) \tag{11.46}$$

Discretization of geometry. In terms of generating the panel corner points $(x_{j=1}, z_{j=1})$, $(x_{j=2}, z_{j=2})$, . . . , $(x_{j=N+1}, z_{j=N+1})$, collocation points $(x_{i=1}, z_{i=1})$, $(x_{i=2}, z_{i=2})$, . . . , $(x_{i=N}, z_{i=N})$, which are placed at the center of each panel, and the normal vectors $\mathbf{n}_i$, the procedure of the previous section can be used (see Fig. 11.18).

Influence coefficients. A possible modification of the boundary condition that will eliminate the zero self-induced effect is to use an internal zero tangential-velocity boundary condition. This is based on Eq. (9.8), which states that inside an enclosed body $\Phi_i^* = \text{const}$. Consequently, the normal and tangential derivatives of the total potential inside the body are zero:

$$\frac{\partial \Phi^*}{\partial n} = \frac{\partial \Phi^*}{\partial l} = 0 \tag{11.47}$$

In this particular case the inner tangential velocity condition will be used and at each panel:

$$(U_\infty + u, W_\infty + w)_i \cdot (\cos \alpha_i, -\sin \alpha_i) = 0 \tag{11.47a}$$

In order to specify this condition at each of the collocation points (which are now at the center of the panel and slightly inside), the tangential velocity component is obtained by using Eq. (11.46). For example, the velocity at a collocation point 1, due to the jth vortex element is

$$(u, w)_{1j} = \text{VOR2DC}\,(\gamma_j, x_1, z_1, x_j, z_j, x_{j+1}, z_{j+1}) \tag{11.48}$$

The influence coefficient a_{ij} is now defined as the velocity component tangent to the surface. Consequently, the contribution of a *unit-strength* singularity element j, at collocation point 1 is

$$a_{1j} = (u, w)_{1j} \cdot (\cos \alpha_1, -\sin \alpha_1)$$

where α_1 is the orientation of the panel (of the collocation point) as shown in Fig. 11.17. The general influence coefficient is then

$$a_{ij} = (u, w)_{ij} \cdot (\cos \alpha_i, -\sin \alpha_i) \tag{11.49}$$

By using this boundary condition the self-induced influence of the panel is

11/20

nonzero, and at the center of the panel, Eqs. (10.42) and (10.43) are recalled,

$$u_p(x, 0\pm) = \pm\frac{\gamma}{2}$$

$$w_p(x, 0\pm) = 0$$

Consequently, when $i = j$ the influence coefficient becomes

$$a_{ii} = -\tfrac{1}{2} \tag{11.50}$$

Establish boundary condition (RHS). The free-stream tangential velocity component RHS_i is found by

$$\text{RHS}_i = -(U_\infty, W_\infty)\cdot(\cos\alpha_i, -\sin\alpha_i) \tag{11.51}$$

Note that in this case the free stream may have an angle of attack. The numerical procedure (using the double "DO loop" routine) for calculating the influence coefficients and the RHS_i vector is the same as for the previous methods.

Solve Equations. Specifying the boundary condition for each $(i = 1 \rightarrow N)$ of the collocation points results in a set of algebraic equations with the unknowns γ_j $(j = 1 \rightarrow N)$. In addition the Kutta condition needs to be specified at the trailing edge:

$$\gamma_1 + \gamma_N = 0 \tag{11.52}$$

But now we have $N + 1$ equations with only N unknowns. Therefore, one of the equations must be deleted (e.g., the kth equation) and by adding the Kutta condition the following matrix equation is obtained:

$$\begin{pmatrix} a_{11} & a_{12} & \cdots & \cdots & a_{1N} \\ a_{21} & a_{22} & \cdots & \cdots & a_{2N} \\ \cdots & \cdots & \cdots & \cdots & \cdots \\ a_{N-1,1} & a_{N-1,2} & \cdots & \cdots & a_{N-1,N} \\ 1 & 0 & 0 & \cdots & 1 \end{pmatrix} \begin{pmatrix} \gamma_1 \\ \gamma_2 \\ \gamma_3 \\ \cdots \\ \gamma_N \end{pmatrix} = \begin{pmatrix} \text{RHS}_1 \\ \text{RHS}_2 \\ \cdots \\ \text{RHS}_{N-1} \\ 0 \end{pmatrix}$$

Caculation of pressures and loads. Once the strength of the vortices γ_j is known, the velocity at each collocation point can be calculated using Eq. (11.48) and the pressure coefficient can be calculated by using Eq. (11.18) (note that the tangential perturbation velocity at each panel is $\gamma_j/2$):

$$C_p = 1 - \left[\frac{Q_\infty \cos(\alpha + \alpha_j) + \gamma_j/2}{Q_\infty}\right]^2 \tag{11.53}$$

The aerodynamic loads can be calculated by adding the pressure coefficient or by using the Kutta–Joukowski theorem. Thus the lift of the jth panel is

$$\Delta L_j = \rho Q_\infty \gamma_j \, \Delta c_j$$

12/20

where Δc_j is the panel length. The total lift and moment are obtained by adding the contribution of each element:

$$L = \sum_{j=1}^{N} \Delta L_j \tag{11.54}$$

$$M_0 = \sum_{j=1}^{N} \Delta L_j (x_j \cos \alpha) \tag{11.55}$$

and the nondimensional coefficients can be calculated by using Eqs. (11.12) and (11.13).

Example 1 Symmetric thick airfoil at angle of attack. The above method is applied to the symmetric airfoil of Section 6.6. The computed pressure distribution is shown by the triangles in Fig. 11.24 and it agrees fairly well with the exact analytical results. The point where the computations disagree is where one equation was deleted. This can easily be corrected by a local smoothing procedure, but the purpose of this example is to highlight this problem. From the practical point of view it is better to use panels with a higher order (e.g., linear) vortex distribution or any of the following methods.

The sample student computer program that was used for this calculation is provided in Appendix D, Program No. 4.



FIGURE 11.24
Chordwise pressure distribution on a symmetric airfoil at angle of attack of 5° using constant-strength vortex panels.

```
C       PROGRAM No. 4: CONSTANT STRENGTH VORTEX
C       ---------------------------------------

C       THIS PROGRAM FINDS THE PRESSURE DISTRIBUTION ON AN ARBITRARY AIRFOIL
C       BY REPRESENTING THE SURFACE AS A FINITE NUMBER OF VORTEX PANELS WITH
C       CONST. STRENGTH (NEUMANN B.C., PROGRAM BY STEVEN YON, 1989).


        REAL EP(400,2),EPT(400,2),PT1(400,2),PT2(400,2)
        REAL CO(400,2),A(400,400),B(400,400),G(400)
        REAL VEL(400),VELT(400),TH(400),DL(400)

        OPEN(8,FILE='CPV.DAT',STATUS='NEW')
        OPEN(9,FILE='AFOIL2.DAT',STATUS='OLD')

        WRITE(6,*) 'ENTER NUMBER OF PANELS'
        READ(5,*) M
        N=M+1
        WRITE(6,*) 'ENTER ANGLE OF ATTACK IN DEGREES'
        READ(5,*) ALPHA
        AL=ALPHA/57.2958

C       READ IN THE PANEL END POINTS
        DO I=1,M+1
         READ(9,*) EPT(I,1), EPT(I,2)
        END DO

C       CONVERT PANELING TO CLOCKWISE
        DO I=1,N
         EP(I,1)=EPT(N-I+1,1)
         EP(I,2)=EPT(N-I+1,2)
        END DO

C       ESTABLISH COORDINATES OF PANEL END POINTS
        DO I=1,M
         PT1(I,1)=EP(I,1)
         PT2(I,1)=EP(I+1,1)
         PT1(I,2)=EP(I,2)
         PT2(I,2)=EP(I+1,2)
        END DO

C       FIND PANEL ANGLES TH(J)
        DO I=1,M
         DZ=PT2(I,2)-PT1(I,2)
         DX=PT2(I,1)-PT1(I,1)
         TH(I)=ATAN2(DZ,DX)
        END DO

C       ESTABLISH COLOCATION POINTS
        DO I=1,M
         CO(I,1)=(PT2(I,1)-PT1(I,1))/2+PT1(I,1)
         CO(I,2)=(PT2(I,2)-PT1(I,2))/2+PT1(I,2)
        END DO

C       ESTABLISH INFLUENCE COEFFICIENTS
        DO I=1,M
         DO J=1,M

C       CONVERT COLOCATION POINT INTO LOCAL PANEL COORDS.
        X2T=PT2(J,1)-PT1(J,1)
        Z2T=PT2(J,2)-PT1(J,2)
        XT=CO(I,1)-PT1(J,1)
        ZT=CO(I,2)-PT1(J,2)

        X2=X2T*COS(TH(J))+Z2T*SIN(TH(J))
        Z2=0
        X=XT*COS(TH(J))+ZT*SIN(TH(J))
        Z=-XT*SIN(TH(J))+ZT*COS(TH(J))
```

```
C       SAVE PANEL LENGTHS FOR LATER USE
        IF(I.EQ.1) THEN
         DL(J)=X2
        END IF

        R1=SQRT(X**2+Z**2)
        R2=SQRT((X-X2)**2+Z**2)

        TH1=ATAN2(Z,X)
        TH2=ATAN2(Z,X-X2)

        IF(I.EQ.J) THEN
         UL=0.5
         WL=0
        ELSE
         UL=0.15916*(TH2-TH1)
         WL=0.15916*LOG(R2/R1)
        END IF

          U=UL*COS(-TH(J))+WL*SIN(-TH(J))
          W=-UL*SIN(-TH(J))+WL*COS(-TH(J))

C       A(I,J) IS THE COMPONENT OF VELOCITY NORMAL TO
C       THE AIRFOIL INDUCED BY THE JTH PANEL AT THE
C       ITH COLOCATION POINT.

        A(I,J)=-U*SIN(TH(I))+W*COS(TH(I))
        B(I,J)=U*COS(TH(I))+W*SIN(TH(I))

         END DO

         A(I,N)=COS(AL)*SIN(TH(I))-SIN(AL)*COS(TH(I))

        END DO

C       REPLACE EQUATION M/4 WITH A KUTTA CONDITION
        DO J=I,M+1
         A(M/4,J)=0
        END DO
         A(M/4,1)=1
         A(M/4,M)=1

C       SOLVE FOR THE SOLUTION VECTOR OF VORTEX STRENGTHS

        CALL MATRX(A,N,G)

C       CONVERT SOURCE STRENGTHS INTO TANGENTIAL
C       VELOCITIES ALONG THE AIRFOIL SURFACE AND CP'S
C       ON EACH OF THE PANELS

 200    CONTINUE

         CL=0
        DO I=1,M
         TEMP=0
        DO J=1,M
         TEMP=TEMP+B(I,J)*G(J)
        END DO
         VEL(I)=TEMP+COS(AL)*COS(TH(I))
     *          +SIN(AL)*SIN(TH(I))
         CL=CL+VEL(I)*DL(I)
        END DO

        WRITE(6,*) 'SMOOTH THE VELOCITY DISTRIBUTION?'
        WRITE(6,*) '1=YES'
        WRITE(6,*) '2=NO'
        READ(5,*) ANS1

        DO I=2,M
         IF(ANS1.EQ.1) THEN
          CP=1-((VEL(I)+VEL(I-1))/2)**2
          WRITE(8,*) PT2(I-1,1),' ,',CP
         ELSE
          CP=1-VEL(I)**2
          WRITE(8,*) CO(I,1),' ,',CP
         END IF
        END DO

        WRITE(6,*) ' '
        WRITE(6,*) 'LIFT COEFFICIENT=', CL

        STOP
        END
```

15/20

11.4.2 Linear-Strength Vortex Method

The constant strength vortex method of Section 11.2.3 posed some difficulties which can be corrected by using the linear strength vortex element. To describe the method let us follow the basic six step procedure:

Selection of singularity element. The linearly varying source distribution shown in Fig. 11.29 includes the same nomenclature that is used for the linearly varying strength vortex panel. The velocity components $(u, w)_p$ in the direction of the panel coordinates were obtained in Sections 10.2 and 10.3 (Eqs. (10.39), (10.40), (10.72), and (10.73)):

$$u_p = \frac{\gamma_0}{2\pi}\left[\tan^{-1}\frac{z}{x-x_2} - \tan^{-1}\frac{z}{x-x_1}\right] + \frac{\gamma_1}{4\pi}\left[z \ln\frac{(x-x_1)^2+z^2}{(x-x_2)^2+z^2} + 2x\left(\tan^{-1}\frac{z}{x-x_2} - \tan^{-1}\frac{z}{x-x_1}\right)\right] \quad \text{panel coordinates} \tag{11.97}$$

$$w_p = -\frac{\gamma_0}{4\pi}\ln\frac{(x-x_1)^2+z^2}{(x-x_2)^2+z^2} - \frac{\gamma_1}{2\pi}\left[\frac{x}{2}\ln\frac{(x-x_1)^2+z^2}{(x-x_2)^2+z^2} + (x_1-x_2) + z\left(\tan^{-1}\frac{z}{x-x_2} - \tan^{-1}\frac{z}{x-x_1}\right)\right] \quad \text{panel coordinates} \tag{11.98}$$

where the subscripts 1, 2 refer to the panel edges j and $j+1$, respectively. As in the case of the linearly varying strength source, it is useful to rearrange these equations in terms of their edge vortex strengths γ_j and γ_{j+1} (see Eqs. (11.88*a* and *b*)) and to use the subscripts j and $j+1$ instead of 1, 2.

$$u_p = \frac{z}{2\pi}\left(\frac{\gamma_{j+1}-\gamma_j}{x_{j+1}-x_j}\right)\ln\frac{r_{j+1}}{r_j} + \frac{\gamma_j(x_{j+1}-x_j)+(\gamma_{j+1}-\gamma_j)(x-x_j)}{2\pi(x_{j+1}-x_j)}(\theta_{j+1}-\theta_j) \quad \text{panel coordinates} \tag{11.99}$$

$$w_p = -\frac{\gamma_j(x_{j+1}-x_j)+(\gamma_{j+1}-\gamma_j)(x-x_j)}{2\pi(x_{j+1}-x_j)}\ln\frac{r_j}{r_{j+1}} + \frac{z}{2\pi}\left(\frac{\gamma_{j+1}-\gamma_j}{x_{j+1}-x_j}\right) \times\left[\frac{(x_{j+1}-x_j)}{z} + (\theta_{j+1}-\theta_j)\right] \quad \text{panel coordinates} \tag{11.100}$$

These two equations combined with the transformation of Eq. (11.23) can be included in a subroutine VOR2DL such that

$$\begin{pmatrix} u, w \\ u^a, w^a \\ u^b, w^b \end{pmatrix}_{ij} = \text{VOR2DL}\,(\gamma_j, \gamma_{j+1}, x_i, z_i, x_j, z_j, x_{j+1}, z_{j+1}) \tag{11.101}$$

where the superscripts $(\ )^a$ and $(\ )^b$ represent the contributions due to the leading and trailing singularity strengths, respectively. For simplicity, this procedure is not repeated here but can be obtained simply by taking all terms multiplied by γ_j in Eqs. (11.99) and (11.100) to produce the $(\ )^a$ component and all terms multiplied by γ_{j+1} to produce the $(\ )^b$ component (as was done in the case of the linearly varying strength source—see Eqs. (11.91) and (11.92)). This decomposition of the velocity components is automatically calculated by Eq. (11.101), and

$$(u, w) = (u^a, w^a) + (u^b, w^b)$$

Discretization of geometry. The panel cornerpoints, collocation points, and normal vectors are computed as in the previous methods.

Influence coefficients. In this phase the zero normal flow boundary condition (Eq. (11.4)) is implemented. For example, the self-induced velocity due to the jth element with a *unit* strength γ_j and γ_{j+1}, at the first collocation point can be obtained by using Eq. (11.101):

$$\begin{pmatrix} u, w \\ u^a, w^a \\ u^b, w^b \end{pmatrix}_{1j} = \text{VOR2DL}\,(\gamma_j = 1,\ \gamma_{j+1} = 1,\ x_1,\ z_1,\ x_j,\ z_j,\ x_{j+1},\ z_{j+1}) \quad (11.102)$$

Similarly to the case of the linearly varying strength source the velocity at each collocation point is influenced by the two edges of the jth panel. Thus, when adding the influence of the $j+1$ panel and on, the local self-induced velocity at the first collocation point will have the form

$$(u, w)_1 = (u^a, w^a)_{11}\gamma_1 + [(u^b, w^b)_{11} + (u^a, w^a)_{12}]\gamma_2 + [(u^b, w^b)_{12} + (u^a, w^a)_{13}]\gamma_3 + \cdots + [(u^b, w^b)_{1,N-1} + (u^a, w^a)_{1N}]\gamma_N + (u^b, w^b)_{1N}\,\gamma_{N+1}$$

This equation can be reduced to a form

$$(u, w)_1 = (u, w)_{11}\gamma_1 + (u, w)_{12}\gamma_2 + \cdots + (u, w)_{1,N+1}\,\gamma_{N+1}$$

such that for the first and last terms

$$(u, w)_{11} = (u^a, w^a)_{11}\,\gamma_1 \quad (11.103a)$$

$$(u, w)_{1,N+1} = (u^b, w^b)_{1N}\,\gamma_{N+1} \quad (11.103b)$$

and for all other terms

$$(u, w)_{1,j} = [(u^b, w^b)_{1,j-1} + (u^a, w^a)_{1,j}]\gamma_j \quad (11.103c)$$

From this point and on the procedure is similar to the linearly varying strength source method. The influence coefficient is calculated when $\gamma_j = 1$ and

$$a_{ij} = (u, w)_{i,j} \cdot \mathbf{n}_i \quad (11.104)$$

For each collocation point there will be $N+1$ such coefficients and unknowns γ_j.

Establish boundary condition (RHS). The free-stream normal velocity component RHS_i is found, as in the case of the discrete vortex (by using Eq. (11.6*a*)):

$$\text{RHS}_i = -(U_\infty, W_\infty) \cdot (\cos\alpha_i, -\sin\alpha_i)$$

Solve equations. Specifying the boundary condition for each ($i = 1 \rightarrow N$) of the collocation points results in N linear algebraic equations with the unknowns γ_j ($j = 1 \rightarrow N+1$). The additional equation can be found by specifying the Kutta condition at the trailing edge:

$$\gamma_1 + \gamma_{N+1} = 0 \quad (11.105)$$

Consequently the set of equations to be solved becomes

$$\begin{pmatrix} a_{11} & a_{12} & \cdots & \cdots & a_{1,N+1} \\ a_{21} & a_{22} & \cdots & \cdots & a_{2,N+1} \\ a_{31} & a_{32} & \cdots & \cdots & a_{3,N+1} \\ \cdots & \cdots & \cdots & \cdots & \cdots \\ a_{N1} & a_{N2} & \cdots & \cdots & a_{N,N+1} \\ 1 & 0 & 0 & \cdots & 1 \end{pmatrix} \begin{pmatrix} \gamma_1 \\ \gamma_2 \\ \gamma_3 \\ \cdots \\ \gamma_N \\ \gamma_{N+1} \end{pmatrix} = \begin{pmatrix} \mathrm{RHS}_1 \\ \mathrm{RHS}_2 \\ \mathrm{RHS}_3 \\ \cdots \\ \mathrm{RHS}_N \\ 0 \end{pmatrix}$$

The above set of algebraic equations has a well-defined diagonal and can be solved for γ_j by using standard methods of linear algebra.

Calculation of pressures and loads. Once the strength of the vortices γ_j is known, the perturbation velocity at each collocation point can be calculated using the results for a vortex distribution (e.g. Eq. (3.147)):

$$Q_{t_j} = (Q_{t_\infty})_j + \frac{\gamma_j + \gamma_{j+1}}{4} \tag{11.106}$$

and the pressure coefficient can be calculated by using Eq. (11.18):

$$C_p = 1 - \frac{Q_t^2}{Q_\infty^2}$$



FIGURE 11.31
Chordwise pressure distribution along a symmetric airfoil (using the linearly varying strength vortex method).

The lift of the panel can be computed from this pressure distribution or by using the Kutta–Joukowski theorem

$$\Delta L_j = \rho Q_\infty \frac{\gamma_j + \gamma_{j+1}}{2} \Delta c_j \tag{11.107}$$

where Δc_j is the panel length. The total lift and moment are obtained by summing the contribution of the individual elements (as in Eqs. (11.54) and (11.55)).

A computer program based on this method is presented in Appendix D (Program No. 6) and the computed pressure coefficients for the airfoil of Section 6.6 at an angle of attack of 5° are presented in Fig. 11.31.

```
C       PROGRAM No. 6: LINEAR STRENGTH VORTEX
C       -------------------------------------

C       THIS PROGRAM FINDS THE PRESSURE DISTRIBUTION ON AN ARBITRARY AIRFOIL
C       BY REPRESENTING THE SURFACE AS A FINITE NUMBER OF VORTEX PANELS WITH
C       LINEAR STRENGTH (NEUMANN B.C., PROGRAM BY STEVEN YON, 1989).

        REAL EP(400,2),EPT(400,2),PT1(400,2),PT2(400,2)
        REAL CO(400,2),A(400,400),B(400,400),G(400)
        REAL TH(400),DL(400)

        OPEN(8,FILE='CPLV.DAT',STATUS='NEW')
        OPEN(9,FILE='AFOIL2.DAT',STATUS='OLD')

        WRITE(6,*) 'ENTER NUMBER OF PANELS'
        READ(5,*) M
        N=M+1
        WRITE(6,*) 'ENTER ANGLE OF ATTACK IN DEGREES'
        READ(5,*) ALPHA
        AL=ALPHA/57.2958

C       READ IN THE PANEL END POINTS
        DO I=1,M+1
         READ(9,*) EPT(I,1), EPT(I,2)
        END DO

        DO I=1,N
         EP(I,1)=EPT(N-I+1,1)
         EP(I,2)=EPT(N-I+1,2)
        END DO

C       ESTABLISH COORDINATES OF PANEL END POINTS
        DO I=1,M
         PT1(I,1)=EP(I,1)
         PT2(I,1)=EP(I+1,1)
         PT1(I,2)=EP(I,2)
         PT2(I,2)=EP(I+1,2)
        END DO
```

```
C        FIND PANEL ANGLES TH(J)
         DO I=1,M
          DZ=PT2(I,2)-PT1(I,2)
          DX=PT2(I,1)-PT1(I,1)
          TH(I)=ATAN2(DZ,DX)
         END DO

C        ESTABLISH COLOCATION POINTS
         DO I=1,M
          CO(I,1)=(PT2(I,1)-PT1(I,1))/2+PT1(I,1)
          CO(I,2)=(PT2(I,2)-PT1(I,2))/2+PT1(I,2)
         END DO

C        ESTABLISH INFLUENCE COEFFICIENTS
         DO I=1,M
          DO J=1,M

C        CONVERT COLOCATION POINT TO LOCAL PANEL COORDS.
         XT=CO(I,1)-PT1(J,1)
         ZT=CO(I,2)-PT1(J,2)
         X2T=PT2(J,1)-PT1(J,1)
         Z2T=PT2(J,2)-PT1(J,2)

         X=XT*COS(TH(J))+ZT*SIN(TH(J))
         Z=-XT*SIN(TH(J))+ZT*COS(TH(J))
         X2=X2T*COS(TH(J))+Z2T*SIN(TH(J))
         Z2=0
C        SAVE PANEL LENGTHS FOR LIFT COEFF. CALC.
         IF(I.EQ.1) THEN
          DL(J)=X2
         END IF
C        FIND R1, R2, TH1, TH2
         R1=SQRT(X**2+Z**2)
         R2=SQRT((X-X2)**2+Z**2)

         TH1=ATAN2(Z,X)
         TH2=ATAN2(Z,X-X2)

C        COMPUTE VELOCITY COMPONANTS AS FUNCTIONS OF
C        GAMMA1 AND GAMMA2. THESE VELOCITIES ARE IN
C        THE JTH REFERENCE FRAME.

         IF(I.EQ.J) THEN
          U1L=-0.5*(X-X2)/(X2)
          U2L=0.5*(X)/(X2)
          W1L=-0.15916
          W2L=0.15916
         ELSE
          U1L=-(Z*LOG(R2/R1)+X*(TH2-TH1)-X2*(TH2-TH1))/
     *         (6.28319*X2)
          U2L=(Z*LOG(R2/R1)+X*(TH2-TH1))/(6.28319*X2)
          W1L=-((X2-Z*(TH2-TH1))-X*LOG(R1/R2)
     *         +X2*LOG(R1/R2))/(6.28319*X2)
          W2L=((X2-Z*(TH2-TH1))-X*LOG(R1/R2))/(6.28319*X2)
         END IF
```

```
C       TRANSFORM THE LOCAL VELOCITIES INTO THE
C       GLOBAL REFERENCE FRAME.

        U1=U1L*COS(-TH(J))+W1L*SIN(-TH(J))
        U2=U2L*COS(-TH(J))+W2L*SIN(-TH(J))
        W1=-U1L*SIN(-TH(J))+W1L*COS(-TH(J))
        W2=-U2L*SIN(-TH(J))+W2L*COS(-TH(J))

C       COMPUTE THE COEFFICIENTS OF GAMMA IN THE
C       INFLUENCE MATRIX.

        IF(J.EQ.1) THEN
         A(I,1)=-U1*SIN(TH(I))+W1*COS(TH(I))
         HOLDA=-U2*SIN(TH(I))+W2*COS(TH(I))
         B(I,1)=U1*COS(TH(I))+W1*SIN(TH(I))
         HOLDB=U2*COS(TH(I))+W2*SIN(TH(I))
        ELSE IF(J.EQ.M) THEN
         A(I,M)=-U1*SIN(TH(I))+W1*COS(TH(I))+HOLDA
         A(I,N)=-U2*SIN(TH(I))+W2*COS(TH(I))
         B(I,M)=U1*COS(TH(I))+W1*SIN(TH(I))+HOLDB
         B(I,N)=U2*COS(TH(I))+W2*SIN(TH(I))
        ELSE
         A(I,J)=-U1*SIN(TH(I))+W1*COS(TH(I))+HOLDA
         HOLDA=-U2*SIN(TH(I))+W2*COS(TH(I))
         B(I,J)=U1*COS(TH(I))+W1*SIN(TH(I))+HOLDB
         HOLDB=U2*COS(TH(I))+W2*SIN(TH(I))
        END IF

        END DO

        A(I,N+1)=COS(AL)*SIN(TH(I))-SIN(AL)*COS(TH(I))

        END DO

C       ADD THE KUTTA CONDITION
         A(N,1)=1
         A(N,N)=1

        IF(M.EQ.10) THEN
         DO I=1,11
          WRITE(6,10) A(I,1),A(I,2),A(I,3),A(I,4),A(I,5),A(I,6),A(I,7),
     *                A(I,8),A(I,9),A(I,10),A(I,11)
         END DO
        END IF

        N=N+1

C       SOLVE FOR THE SOLUTION VECTOR OF VORTEX STRENGTHS

        CALL MATRX(A,N,G)

C       CONVERT VORTEX STRENGTHS INTO TANGENTIAL
C       VELOCITIES ALONG THE AIRFOIL SURFACE AND CP'S
C       ON EACH OF THE PANELS.

 200    CONTINUE

        N=M+1
        CL=0

        DO I=1,M
          VEL=0
         DO J=1,N
          VEL=VEL+B(I,J)*G(J)
         END DO
          V=VEL+COS(AL)*COS(TH(I))+SIN(AL)*SIN(TH(I))
          CL=CL+V*DL(I)
          CP=1-V**2
          WRITE(8,*) CO(I,1),' ,',CP
         END DO

        WRITE(6,*) ' '
        WRITE(6,*) 'LIFT COEFFICIENT=',CL

        STOP

 10     FORMAT(/,F6.2,1X,F5.2,1X,F5.2,1X,F5.2,1X,F5.2,1X,F5.2,1X,F5.2,
     *         1X,F5.2,1X,F5.2,1X,F5.2,1X,F5.2)

        END
```